
1. Introducción

Un metamaterial es un sistema arti�cial formado por partículas intercala-
das de dos o mas materiales ordinarios. Su estudio ha adquirido importancia
pues sus propiedades ópticas pueden diferir sustancialmente de las propie-
dades de cada una de sus componentes, e incluso, pueden ser propiedades
exóticas que no existen en los materiales ordinarios. El propósito de estas
notas es estudiar las propiedades de uno de los metamateriales más simples,
una superred formada por la alternancia de dos películas. Cuando una de las
películas es metálica y la otra aislante y bajo ciertas aproximaciones, el me-
tamaterial puede tratarse como un medio homogéneo pero con propiedades
dieléctricas anisotrópicas, como si fuera un cristal uniaxial. En particular, a
bajas frecuencias la respuesta es de caracter metálico en la dirección normal
a la super�cie y aislante en la dirección perpendicular. Por ello, la relación de
dispersión de los modos propios correspondientes a los rayos extraordinarios
de los materiales uniaxiales, es de caracter hiperbólico en lugar de elíptico.
Esto tiene consecuencias interesantes como el corresponder a una densidad
in�nita de estados fotónicos. En las presentes notas desarrollaremos la teoría
de medio efectivo que describe este metamaterial, obtendremos expresiones
para sus modos propios electromagnéticos de bulto y de super�cie, desarro-
llaremos códigos computacionales para su cálculo y haremos un análisis de
los resultados para un modelo sencillo. 1

1Estas notas fueron preparadas usando el modo Org del editor emacs y exportadas a

LaTex para generar el archivo pdf. Los códigos fueron programados en el lenguaje Perl y

su ampliación numérica Perl Data Language.

2. Medio efectivo

2.1. Teoría

Consideremos una superred metal-aislante formada por películas alter-
nadas a y b de anchos da y db con funciones dieléctricas εa y εb. Me interesa
el caso en que uno de los dos materiales, digamos a es un metal y el otro,
digamos b es un dieléctrico. Me interesa calcular el plasmón de super�cie del
sistema. Como una primera aproximación, estudiaremos el caso en que da y
db son muy pequeños.

Cuando las películas que conforman al sistema son muy delgadas compa-
radas con otras distancias relevantes como podrían ser la longitud de onda
del campo o la distancia de penetración en cada uno de los materiales, la
respuesta macroscópica del sistema puede hallarse de manera muy sencilla
haciendo las siguientes consideraciones: Las proyecciones E‖ y D⊥ del campo
eléctrico y del campo magnético son contínuas al atravesar cada interfase, y
estos campos son prácticamente constantes al atravesar cada película delga-
da. Por lo tanto puedo identi�car estos campos como campos macroscópicos,
i.e.,

E‖ = EM‖ , D⊥ = EM⊥ . (1)

Por lo tanto, podemos escribir

D‖ = ε(z)E‖(z) = ε(z)EM‖ , E⊥ = ε−1(z)D⊥(z) =
1

ε(z)
DM
⊥ , (2)

ignorando la variación espacial de los campo macroscópicos a lo largo de un
periodo, donde

ε(z) =

{
εa, en las capas a

εb, en las capas b
(3)

Promediando las ecs. (2) obtenemos los campos macroscópicos

DM
‖ = 〈D〉 = 〈ε(z)EM‖ 〉 = 〈ε(z)〉E

M
‖ ,

EM⊥ =

〈
1

ε(z)
DM
⊥

〉
=

〈
1

ε(z)

〉
DM
⊥ ,

(4)

de donde podemos identi�car la respuesta macroscópica

εM‖ = 〈ε〉 = faεa + fbεb,
1

εM⊥
=

〈
1

ε(z)

〉
=
fa
εa

+
fb
εb
, (5)

donde fa = da/d y fb = db/d son las fracciones de llenado de los materiales
a y b.

Como ejemplo, supongamos que a es un metal descrito por el modelo de
Drude

εa = 1−
ω2
p

ω2
, (6)

y que b es un dieléctrico con una función dieléctrica constante. Entonces,

εM‖ = fa + fbεb − fa
ω2
p

ω2

εM⊥ =
εb

faεb + fb
+
faε

2
bω

2
p/(faεb + fb)

2

fb
faεb+fb

ω2
p − ω2

.
(7)

Puedo reescribir estas ecuaciones como

εM‖ = ε∞‖ −
ω2
D‖

ω2
, εM⊥ = ε∞⊥ +

ω2
D⊥

ω2
0⊥ − ω2

, (8)

donde
ε∞‖ = fa + fbεb, ε∞⊥ =

εb
faεb + fb

, (9)

son las respuestas asintóticas a alta frecuencia,

ω2
D‖ = faω

2
p, ω2

D⊥ =
faε

2
bω

2
p

(faεb + fb)2
(10)

son análogos a la frecuencia de Drude
√
4πne2/m, con n la densidad de

portadores de masa m y carga e en el modelo lorentziano, y

ω2
0⊥ =

fb
faεb + fb

ω2
p, (11)

es una frecuencia de resonancia.
Notamos que la respuesta en la dirección paralela es como la respuesta de

un conductor, pero con una respuesta de alta frecuencia distinta de 1, debido
a la contribución del aislante, y con una frecuencia de Drude disminuida
debido a la fracción de llenado metálica. El cero de esta respuesta se halla
en

ωL‖ =

√
fa

fa + fbεb
ωp. (12)

Comparando con la ec. (11) notamos que para la fracción de llenado fa =
fb = 0,5 el polo de εM⊥ coincide con el cero de εM‖ .

Por otro lado, la respuesta en la dirección perpendicular es como la de un
aislante pero con una resonancia en ω0⊥. El sistema es aislante a bajas fre-
cuencias pues los dieléctricos impiden el paso de carga en la dirección normal
a las interfaces, pero tiene una resonancia asociada al movimiento colectivo
de carga en la dirección normal dentro de la película metálica: electrones
acumulados en una interface son acelerados hacia la interface opuesta hasta
que su acumulación en esta frena su movimiento e inicia un movimiento hacia
la interface original, produciendo un movimiento. Para una película aislada
en el vacío, la frecuencia natural sería la frecuencia de plasma ωp, pero en
este sistema se ve modi�cada por polarización del dieléctrico. Curiosamente,
εM⊥ = 0 en la frecuencia ωL⊥ = ωp, al igual que el metal homogéneo.

En resumen, una superred de periodo pequeño se comporta como un
sistema homogéneo pero anisotrópico, con respuestas diferentes a lo largo de
las interfaces y en la dirección perpendicular. En el caso de una superred
conductor-aislante, la respuesta es como un conductor de menor densidad a
lo largo de la super�cie y como un aislante con una resonancia lorentziana
en la dirección perpendicular.

2.2. Código

A continuación presento un código para calcular las funciones respuesta
discutidas arriba. Iniciamos cargando los encabezados usuales y los paquetes
requeridos.

use v5.12;

use warnings;

use Getopt::Long;

use IO::Prompter;

use PDL;

use PDL::NiceSlice;

use PDL::Graphics::Gnuplot;

Figura 1: paquetes
Declara e inicializa los parámetros,

my $wp; # frecuencia de plasma medio a

my $epsb; # función dieléctrica del medio b

my $fa; # fracción de llenado del metal

my ($wmin, $wmax, $Nw); # frecuencias mínima, máxima y número

my ($e0, $e1); # Minimum and maximum epsilons to plot

my $options=q(

4



'wp=f'=>\$wp, # frecuencia de plasma medio a

'epsb=f'=>\$epsb, # función dieléctrica del medio b

'fa=f'=>\$fa, # fracción de llenado del metal

'wmin=f'=>\$wmin, 'wmax=f'=>\$wmax, 'Nw=i'=>\$Nw, # frecuencias mínima, máxima y número

'e0=f'=>\$e0, 'e1=f'=>\$e1, # Minimum and maximum epsilons to plot

);

my %options=(eval $options);

die "Error en definición de opciones; $@" if $@;

GetOptions(%options) or usage($options, "Opciones erróneas");

usage($options, "Faltan parámetros")

unless List::Util::all {defined $_} ($wp, $epsb, $fa, $wmin, $wmax, $Nw);

usage($options, "La fracción de llenado debe estar entre 0 y 1") unless 0<=$fa<=1;

my $fb=1-$fa; # fraccion de llenado del aislante.

my $w=zeroes($Nw)->xlinvals($wmin,$wmax); # Arreglo de frecuencias

my $epsa=1-$wp**2/$w**2; # Drude

Figura 2: inicializa
Necesitamos una rutina para explicar el uso si hubiera un error.

sub usage {

my ($options, $message)=@_;

say $message if defined $message;

say $options;

exit 1;

}

Figura 3: uso
Calcula la respuesta macroscópica,

# calcula la respuesta directamente

my $eM_par=$fa*$epsa+$fb*$epsb;

my $eM_perp=1/($fa/$epsa+$fb/$epsb);

# # verifica calculándola mediante las expresiones lorentzianas

# my $e_par_inf=$fa+$fb*$epsb;

# my $e_perp_inf=$epsb/($fa*$epsb+$fb);

# my $wD2_par=$fa*$wp**2;

# my $wD2_perp=$fa*($epsb*$wp/($fa*$epsb+$fb))**2;

# my $w02_perp=$fb/($fa*$epsb+$fb)*$wp**2;

# my $eM_par1=$e_par_inf-$wD2_par/$w**2;

# my $eM_perp1=$e_perp_inf+$wD2_perp/($w02_perp-$w**2);

Figura 4: calcula
Gra�ca los resultados

my $gw=gpwin();

my $S="{/Symbol %s}";

$gw->plot({title=>sprintf("Respuesta macroscópica ${S}_p=%s ${S}_b=%s f_a=%s f_b=%s",

'w', $wp, 'e', $epsb, $fa, $fb),

ylabel=>'{/Symbol e}^M', xlabel=>'Frecuencia {/Symbol w}',

yrange=>[$e0,$e1], xzeroaxis=>1},

{legend=>'par', with=>'lines',}, $w, $eM_par,

{legend=>'perp', with=>'lines',}, $w, $eM_perp,

# {legend=>'par alt', with=>'lines',}, $w, $eM_par1,

# {legend=>'perp alt', with=>'lines',}, $w, $eM_perp1,

);

prompt -single, -void, "Listo?";

Figura 5: gra�ca
Finalmente, coloca todo junto.

#Calcula la respuesta macroscópica de una superred metal aislante.

<<paquetes>>

<<inicializa>>

<<calcula>>

<<grafica>>

<<uso>>

Figura 6: todo

2.3. Resultados

Corrimos el programa 6 como en la �g. 7, normalizando las frecuencias a
la frecuencia de plasma indicando una función dieléctrica εb = 2, y haciendo
cálculos para 1000 frecuencias entre .1 y 3.5, empleando momentos angulares
l = 0, 1, . . . 10.

./epsilonM.pl -wp=1 -epsb=2 -fa= .5 -wmin=.01 -wmax=2 -Nw=100\

-e0=-40 -e1=40

./epsilonM.pl -wp=1 -epsb=2 -fa=.25 -wmin=.01 -wmax=2 -Nw=100\

-e0=-40 -e1=40

./epsilonM.pl -wp=1 -epsb=2 -fa=.75 -wmin=.01 -wmax=2 -Nw=100\

-e0=-40 -e1=40

Figura 7: Ejecución del programa 6.
En la �g. 8 mostramos las respuestas macroscópicas εM‖ y εM⊥ para una

superred hecha de un metal de Drude con ωp = 1 y un dieléctrico con εb =
2 con fracciones de llenado simétricas fa = fb = 0,5. Veri�camos que la
respuesta es metálica en la dirección paralela y dieléctrica en la dirección
perpendicular. En esta última hay una resonancia que coincide con el cero
de la respuesta paralela, y hay un cero en ωp.

Figura 8: εM‖ y εM⊥ para una superred metal-aislante con ωp = 1, εb = 2 y
con fracciones de llenado fa = fb = 0,5.

La �g. 9 es análoga a la �g. 8 pero con una fracción de llenado menor
para el metal y mayor para el dieléctrico, fa = 0,25 y fb = 0,75. En este
caso hay dos regiones, w < 0,38ωp y ω > 0,79ωp en las que una o la otra
respuesta es negativa, mientras que en la región 0,38ωp < ω < 0,79ωp ambas
respuestas son positivas. La �g. 10 es análoga a las dos anteriores pero con
una fracción de llenado mayor para el metal, fa = 0,75 y fb = 0,25. En este
caso, para w < 0,38ωp una de las respuestas es negativa y la otra positiva,
y para ω > ωp ambas son positivas, pero para 0,38ωp < ω < ωp ambas
funciones dieléctricas son negativas.

Figura 9: εM‖ y εM⊥ como en la �g. 8 pero fa0,25 y fb = 0,25.

Figura 10: εM‖ y εM⊥ como en la �g. 8 pero fa = 0,75 y fb = 0,25.

3. Modos propios

En la sección anterior vimos que una superred con un periodo muy pe-
queño puede describirse como un material homogéneo anisotrópico con una
respuesta dieléctrica paralela εM‖ y una respuesta perpendicular εM⊥ , en ge-
neral distintas.

3.1. Bulto

Consideremos una onda con polarización TM que se propaga con frecuen-
cia ω y con vector de onda k. La ecuación de onda para el campo eléctrico
puede escribirse como

k × (k × E) = −q2D, (13)

donde abreviamos q = ω/c y omitimos el índice M para denotar macros-
cópico por ya no ser necesario. Escribiendo D‖ = ε‖E‖ y D⊥ = ε⊥E⊥ y
desarrollando el triple producto vectorial, podemos reescribir la ec. (13) co-
mo la ecuación matricial(

k2⊥ − ε‖q2 −k⊥k‖
−k⊥k‖ k2‖ − ε⊥q

2

)(
E‖
E⊥

)
= 0, (14)

Los modos propios de la superred están dados entonces por los ceros del
determinante,

det

(
k2⊥ − ε‖q2 −k⊥k‖
−k⊥k‖ k2‖ − ε⊥q

2

)
= 0, (15)

lo cual se puede escribir como

k2⊥
ε‖

+
k2‖

ε⊥
= q2. (16)

3.2. Super�cie

Consideremos una superred semiin�nita, cortada a lo largo de sus inter-
faces. Consideremos una onda que incide desde el vacío con un vector de
onda ki y frecuencia ω = qc, y llamemos Q a su proyección sobre la inter-
face Q = ki‖, la cual, por conservación del ímpetu es igual a la proyección

kt‖ = Q del vector de onda kt de la onda transmitida. Las componentes
perpendiculares están dadas entonces por

(ki⊥)
2 = q2 −Q2, (kt⊥)

2 = ε‖q
2 − ε‖

Q2

ε⊥
. (17)

A partir de la ecuación de Ampére-Maxwell podemos obtener la impedancia
super�cial para las ondas incidente y transmitida

Zi =
ki⊥c

ω
, Zt =

kt⊥c

ε‖ω
. (18)

De aquí podemos obtener la amplitud de re�exión rp = (Zi−Zt)/(Zi+Zt) y
los plasmones polaritones de super�cie, dados intrínsicamente por Zi+Zt =
0. Sustituyendo la ec. (18) tenemos

kt⊥ = −ε‖ki⊥. (19)

Elevando al cuadrado ambos lados de esta ecuación, sustituyendo la ec. (17)
y despejando Q2 obtenemos

Q2 =
(ε‖ − 1)ε⊥

ε‖ε⊥ − 1

ω2

c2
. (20)

Podemos veri�car que en el caso isotrópico ε‖ = ε⊥ = ε esta ecuación se
reduce al resultado usual Q2 = (ε/(ε+ 1)ω2/c2.

Es importante notar que al elevar al cuadrado la ec. (19) aparecen en la
misma soluciones de la ecuación kt⊥ = ε‖k

i
⊥ que no describen a un plasmón-

polaritón de super�cie, sino más bien, a la condición de Brewster de tener
re�ectancia cero.

Para el límite no retardado, cuando Q → ∞, podemos obtener la fre-
cuencia asintótica a partir del cero del denominador de la ec. (20), ε‖ε⊥ = 1.
Despejando ω obtenemos

ω2
spp =

2εbfa − fb + ε2bfb −
√
f2b + ε4bf

2
b + ε2b(4f

2
a − 2f2b )

(ε2b − 1)fb

ω2
p

2
. (21)

3.3. Código. Modos de bulto.

A continuación elaboraremos un programa para gra�car la relación de
dispersión en el bulto de una superred metal-aislante.

Empezamos con encabezados y paquetes, para lo cual tomaremos pres-
tado el de la �g. 1. Luego declaramos e inicializamos los parámetros.

my $wp; # frecuencia de plasma medio a

my $epsb; # función dieléctrica del medio b

my $fa; # fracción de llenado del metal

my ($wmin, $wmax, $Nw); # frecuencias mínima, máxima y número

my ($Qmin, $Qmax, $NQ); # vector de onda paralelo mínimo y máximo

my ($k0, $k1); # rango de k's para graficar

my $options=q(

'wp=f'=>\$wp, # frecuencia de plasma medio a

'epsb=f'=>\$epsb, # función dieléctrica del medio b

'fa=f'=>\$fa, # fracción de llenado del metal

'wmin=f'=>\$wmin, 'wmax=f'=>\$wmax, 'Nw=i'=>\$Nw, # frecuencias mínima, máxima y número

'Qmin=f'=>\$Qmin, 'Qmax=f'=>\$Qmax, 'NQ=f'=>\$NQ, # vector de onda paralelo mínimo y máximo

'k0=f'=>\$k0, 'k1=f'=>\$k1, # rango de k's para graficar

);

my %options=(eval $options);

die "Error en definición de opciones; $@" if $@;

GetOptions(%options) or usage($options, "Opciones erróneas");

usage($options, "Faltan parámetros")

unless List::Util::all {defined $_} ($wp, $epsb, $fa, $wmin, $wmax, $Nw, $Qmin, $Qmax, $NQ);

usage($options, "La fracción de llenado debe estar entre 0 y 1") unless 0<=$fa<=1;

my $fb=1-$fa; # fraccion de llenado del aislante.

my $w=zeroes($Nw)->xlinvals($wmin,$wmax); # Arreglo de frecuencias

my $c=1;

my $q=$w/$c; # números de onda en el vacío. Supongo c=1

my $Q=$NQ>=2?zeroes($NQ)->xlinvals($Qmin, $Qmax)

:pdl([$Qmin]); # Arreglo de vectores paralelos

my $epsa=1-$wp**2/$w**2; # Drude

Figura 11: inicializa1
Calcula la respuesta macroscópica reciclando el código de la �g. 4, y la

emplea para obtener la relación de dispersión del bulto usando las ec. (16).

my $k_perp=sqrt(($eM_par*($q**2-$Q->dummy(0)**2/$eM_perp))->r2C)->re;

Figura 12: reldisB
Gra�ca el resultado.

my $gw=gpwin();

my $S="{/Symbol %s}";

$gw->plot({title=>sprintf("Relación de dispersión normal ${S}_p=%s ${S}_b=%s f_a=%s f_b=%s Q=%s..%s",

'w', $wp, 'e', $epsb, $fa, $fb, $Qmin, $Qmax),

xlabel=>'Vector de onda k^M_{perp}', ylabel=>'Frecuencia {/Symbol w}',

xrange=>[$k0,$k1],

xzeroaxis=>1},

with=>'lines', $k_perp, $w,);

prompt -single, -void, "Listo?";

Figura 13: gra�ca1
Finalmente, coloca todo junto.

#Calcula la respuesta macroscópica de una superred metal aislante.

<<paquetes>>

<<inicializa1>>

<<calcula>>

<<reldisB>>

<<grafica1>>

<<uso>>

Figura 14: todo

3.4. Resultados: Relación de dispersión de bulto

Podemos correr el programa como en la �g. 15.

./reldisB.pl -wp=1 -epsb=2 -fa=.5 -wmin=.01 -wmax=3 -Nw=100 -Qmin 0 \

-Qmax=1 -NQ=10 -k1 4

./reldisB.pl -wp=1 -epsb=2 -fa=.25 -wmin=.01 -wmax=3 -Nw=100 -Qmin 0 \

-Qmax=1 -NQ=10 -k1 4

./reldisB.pl -wp=1 -epsb=2 -fa=.75 -wmin=.01 -wmax=3 -Nw=100 -Qmin 0 \

-Qmax=1 -NQ=10 -k1 4

Figura 15: corrida1
Para la superred simétrica el resultado se ve en la �g. 16 para distintos

valores de Q. Para Q = 0 hay propagación empezando en ωL‖ ≈ 0,58ωp
y continúa ascendiendo como una hipérbola. Al aumentar Q aparece un
cambio cualitativo. El modo tiene una asíntota Q → ∞ en ω = ωp y tras
una pequeña brecha prohibida reaparece como una nueva hipérbola con una
frecuencia ω > ωp para k⊥ = 0. Además aparece un nuevo modo de frecuencia
baja con velocidad de grupo negativa. Para ilustrar este comportamiento, en
la �g. 17 mostramos sólo una de las curvas de la �g. 16.

Podemos entender este resultado a partir de la �g. 8 y la relación de
dispersión (16). Como en este caso una función dieléctrica es negativa y la
otra positiva para ω < ωp, por lo que la dispersión es hiperbólica, pero los
papeles de ε‖ y ε⊥ se intercambian en la resonancia ω0⊥. Por ello hay dos

Figura 16: Relación de dispersión k⊥ vs. ω para diversos valores del vector
de onda paralelo Q = 0.,1.

Figura 17: Relación de dispersión k⊥ vs. ω para un solo valor de Q = 1.

soluciones abajo de ωp. Arriba de ωp ambas funciones respuesta son positivas
y el material se comporta como un medio uniaxial transperente usual.

En la �g. 18 mostramos la relación de dispersión de los modos de bulto
para el caso fa = 0,25. El resultado es similar al de la �g. 16 aunque ahora
ωL‖ no coincide con ω0⊥ y entre ambas aparece otra región en que ambas
funciones respuesta son positivas. En este caso, la frecuencia de las tres ra-
mas incrementan su frecuencia el límite k⊥ → 0 al incrementar Q, y la rama
de enmedio no toma un valor �jo como la de la �g. 16. Sin embargo, en ω0⊥
hay un polo de εM⊥ , por lo que todas las curvas se degeneran en esta frecuen-
cia. Arriba de ω0⊥, ε

M
⊥ < 0 < εM‖ y k⊥ diverge conforme nos acercamos a

ωL⊥ = ωp. Finalmente, arriba de esta frecuencia ambas funciones respuesta
son positivas y la dispersión se vuelve similar a la de la �gura 16 Notamos
que la banda inferior para Q's grandes se traslapa con la banda intermedia
para Q's pequeñas.

En la �g. 19 mostramos la relación de dispersión de los modos de bulto
para el caso fa = 0,75.

El resultado es de nuevo similar al de la �g. 16. Ahora ωL‖ > ω0⊥ y
entre ambas hay una región donde ambas respuestas son negativas y no

Figura 18: Relación de dispersión k⊥ vs. ω como en la �g. 16 pero para una
fracción de llenado fa = 0,25.

Figura 19: Relación de dispersión k⊥ vs. ω como en la �g. 16 pero para una
fracción de llenado fa = 0,75.

puede haber propagación. Por ello, en este caso no se tocan y menos aún se
traslapan las dos bandas inferiores. Como en la �g. 16, el límite de ω cuando
k ⊥→ 0 en la banda de enmedio no depende de Q.

3.5. Código: Curvas de nivel

El comportamiento de las relaciones de dispersión mostrado en la sección
anterior puede entenderse a partir del estudio de sus curvas de nivel, i.e., las
líneas en el espacio k‖, k⊥ en las que ω es constante. Para gra�carlas hacemos
el siguiente código. Tomamos prestada el código la �gura 1. Mostramos en
la �gura el código para declarar e inicializar variables es.

my $wp; # frecuencia de plasma medio a

my $epsb; # función dieléctrica del medio b

my $fa; # fracción de llenado del metal

my ($wmin, $wmax, $Nw); # frecuencias mínima, máxima y número

my ($kmax, $Nk); # vector de onda máximo, número de k's

my $options=q(

'wp=f'=>\$wp, # frecuencia de plasma medio a

'epsb=f'=>\$epsb, # función dieléctrica del medio b

'fa=f'=>\$fa, # fracción de llenado del metal

'wmin=f'=>\$wmin, 'wmax=f'=>\$wmax, 'Nw=i'=>\$Nw, # frecuencias mínima, máxima y número

'kmax=f'=>\$kmax, 'Nk=i'=>\$Nk, # vector de onda máximo, número de k's

);

my %options=(eval $options);

die "Error en definición de opciones; $@" if $@;

GetOptions(%options) or usage($options, "Opciones erróneas");

usage($options, "Faltan parámetros")

unless List::Util::all {defined $_} ($wp, $epsb, $fa, $wmin, $wmax, $Nw, $kmax, $Nk);

usage($options, "La fracción de llenado debe estar entre 0 y 1") unless 0<=$fa<=1;

my $fb=1-$fa; # fraccion de llenado del aislante.

my $w=$Nw>=2?zeroes($Nw)->xlinvals($wmin,$wmax) # Arreglo de frecuencias

:pdl([$wmin]);

my $Q=zeroes($Nk)->xlinvals(-$kmax,$kmax); # Arreglo de Q's

my $c=1;

my $q=$w/$c; # números de onda en el vacío. Supongo c=1

my $epsa=1-$wp**2/$w**2; # Drude

Figura 20: inicializa3

Calcula la respuesta macroscópica reciclando el código de la �g. 4, y
empléala para despejar k⊥ a partir de la relación de dispersión del bulto
usando las ec. (17).

my $k_perp=sqrt($eM_par*($q**2-$Q->dummy(0)**2/$eM_perp))->mv(1,0); # [Q,w]

Figura 21: kperp
Gra�ca el resultado.

my $gw=gpwin();

my $k_perp_full=append(-$k_perp(-1:0),$k_perp);

my $Q_full=append($Q(-1:0),$Q);

my $S="{/Symbol %s}";

$gw->plot({title=>sprintf(

"Curvas de nivel rel. de dispersión normal ${S}_p=%s ${S}_b=%s f_a=%s f_b=%s"

. "${S}_{min}=%s ${S}_{max}=%s",

'w', $wp, 'e', $epsb, $fa, $fb, 'w', $wmin, 'w', $wmax),

xlabel=>'Vector de onda k_{par}',

ylabel=>'Vector de onda k_{perp}',

xrange=>[-$kmax,$kmax],

yrange=>[-$kmax,$kmax],

justify=>1},

with=>'lines', $Q_full, $k_perp_full);

prompt -single, -void, "Listo?";

Figura 22: gra�ca3
Finalmente, coloca todo junto.

#Calcula la respuesta macroscópica de una superred metal aislante.

<<paquetes>>

<<inicializa3>>

<<calcula>>

<<kperp>>

<<grafica3>>

<<uso>>

Figura 23: todo3

3.6. Resultados: Cuvas de nivel

Corrimos el código como en la �g. 24

./nivel.pl -wp=1 -epsb=2 -fa=.5 -wmin=.1 -wmax=.55 -Nw=10 -kmax=4 -Nk=1000

./nivel.pl -wp=1 -epsb=2 -fa=.5 -wmin=.6 -wmax=.999 -Nw=11 -kmax=4 -Nk=1000

./nivel.pl -wp=1 -epsb=2 -fa=.5 -wmin=1.01 -wmax=3 -Nw=10 -kmax=4 -Nk=1000

./nivel.pl -wp=1 -epsb=2 -fa=.25 -wmin=.1 -wmax=.35 -Nw=10 -kmax=4 -Nk=1000

./nivel.pl -wp=1 -epsb=2 -fa=.25 -wmin=.4 -wmax=.76 -Nw=10 -kmax=4 -Nk=1000

./nivel.pl -wp=1 -epsb=2 -fa=.25 -wmin=.8 -wmax=.999 -Nw=10 -kmax=4 -Nk=1000

./nivel.pl -wp=1 -epsb=2 -fa=.25 -wmin=1.1 -wmax=3 -Nw=10 -kmax=4 -Nk=1000

./nivel.pl -wp=1 -epsb=2 -fa=.75 -wmin=.78 -wmax=.999 -Nw=10 -kmax=4 -Nk=1000

./nivel.pl -wp=1 -epsb=2 -fa=.75 -wmin=1.1 -wmax=3 -Nw=10 -kmax=4 -Nk=1000

Figura 24: corrida3
En la �gura 25 mostramos algunas curvas de nivel k(ω) correspondientes

a frecuencias bajas omega < ωL‖ = ω0⊥. Notamos que las curvas son hipér-
bolas que se abren a derecha e izquierda. Esto se debe a que en esta región

Figura 25: Curvas de nivel k(ω) para algunas frecuencias bajas ω < ωL‖ =
ω0⊥ para el mismo sistema que en la �gura 8, ωp = 1, εb = 2, fa = fb = 0,5.

ε‖ < 0 mientras que ε⊥ > 0. Conforme la frecuencia disminuye, las curvas
tienen asíntotas más verticales, y por tanto k⊥ toma valores mayores, como
muestra la parte inferior de la �gura 16.

Las hipérbolas se convierten en líneas horizontales conforme nos acerca-
mos a ωL‖ = ω0⊥ ≈ 0,578, donde ambas funciones cambian de signo y las
curvas de nivel cambian a hipérbolas que abren hacia arriba y abajo, como
ilustra la �gura 26. Las de frecuencia más baja son horizontales, y por eso
la frecuencia correspondiente a k⊥ ≈ 0 no depende del valor de $k_\|, como
muestra la parte intermedia de la �g. 16. Conforme aumenta la frecuencia

Figura 26: Curvas de nivel como en la �gura 25 pero para frecuencias
$ωL\|=ω0⊥<ω<ωp=ωL⊥.

la relación de dispersión converge a dos líneas verticales en ω = ωp, por lo
que k⊥ diverge. Pasando dicha frecuencia, ε⊥ se vuelve positiva y las curvas
de nivel se convierten en elipses, como muestra la �g. 27 , correspondiente
la parte alta de la �gura 16. Las �gura 28 es análoga a la �g. 25 pero para
una fracción de llenado metálica menor, fa = 0,25. Esta �gura es cualitati-
vamente similar a la �g. 25, con hipérbolas que abren a derecha e izquierda
que se degeneran en una línea recta vertical a bajas frecuencias y en dos
líneas rectas horizontales alrededor de ωL‖. Estas curvas corresponden a la
parte baja de la �gura 18. Para frecuencias arriba de ωL‖ ambas respuestas
se vuelven positivas y las curvas de nivel se convierten en elipses, como en
la �g. 29, correspondiente a la parte intermedia de la �g. 18. Conforme nos

Figura 27: Curvas de nivel como en la �gura 25 pero frecuencias ω > ωp.

acercamos al polo ω0⊥ de εM⊥ estas elipses degeneran en dos líneas horizon-
tales, que corresponden al punto donde se cruzan todas las curvas en la �g.
18. En la región ω0⊥ < ω < ωp = ωL⊥, ε⊥ < 0 < εM‖ , las curvas de nivel son
hipérbolas como en la �gura 26 y la relación de dispersión en la �g. 18 es
similar a la región intermedia de la �gura 16.

22



Figura 28: Curvas de nivel, como en la �g. 25, pero para una fracción de
llenado fa = 0,25 y para frecuencias ω < ωL‖.

Figura 29: Curvas de nivel como en la �g. 28 pero a frecuencias ωL‖ < ω <
ω0⊥.

Finalmente, para ω > ωp = ωL⊥ ambas respuestas son positivas, las curvas
de nivel son elipses, como muestra la �g. 30 y la relación de dispersión en la
parte superior de la �g. 18 es cualitativamente similar a la correspondiente
de la �g. 16.

Por último, en las �guras 31, 32, y 33 mostramos las curvas de nivel para
una fracción de llenado metálica relativamente alta fa = 0,75. Estas curvas
corresponden a las regiones ω < ω0⊥, ωL‖ < ω < ωL⊥ = ωp y ωL⊥ = ωp < ω
y las curvas de nivel son similares a las de las �guras 25, 26 y 27. Sin embargo,
en la región ω0⊥ω < ωL‖ ambas funciones dieléctricas son negativas y aparece
una brecha prohibida, como podemos apreciar en la �g. 19, y que no aparece
en los casos anteriores.

3.7. Código: Modos de super�cie

A continuación elaboraremos un programa para gra�car la relación de
dispersión de los plasmones polaritones de super�cie de una superred metal-
aislante.

De nuevo, tomamos los encabezados y paquetes prestados de la �g. 1.
Luego declaramos e inicializamos los parámetros.

my $wp; # frecuencia de plasma medio a

my $epsb; # función dieléctrica del medio b

Figura 30: Curvas de nivel como en la �g. 28 pero a frecuencias ωp = ωL⊥ <
ω.

Figura 31: Curvas de nivel, como en la �g. 25, pero para una fracción de
llenado fa = 0,75 y para frecuencias ω < ω0⊥.

Figura 32: Curvas de nivel como en la �g. 31 pero a frecuencias ωL‖ < ω <
ωL⊥ = ωp.

Figura 33: Curvas de nivel como en la �g. 31 pero a frecuencias ωp = ωL⊥ <
ω.

my $fa; # fracción de llenado del metal

my ($wmin, $wmax, $Nw); # frecuencias mínima, máxima y número

my $Q1; #máximo vector de onda

my $options=q(

'wp=f'=>\$wp, # frecuencia de plasma medio a

'epsb=f'=>\$epsb, # función dieléctrica del medio b

'fa=f'=>\$fa, # fracción de llenado del metal

'wmin=f'=>\$wmin, 'wmax=f'=>\$wmax, 'Nw=i'=>\$Nw, # frecuencias mínima, máxima y número

'Q1=f'=>\$Q1, # máximo vector de onda

);

my %options=(eval $options);

die "Error en definición de opciones; $@" if $@;

GetOptions(%options) or usage($options, "Opciones erróneas");

usage($options, "Faltan parámetros")

unless List::Util::all {defined $_} ($wp, $epsb, $fa, $wmin, $wmax, $Nw);

usage($options, "La fracción de llenado debe estar entre 0 y 1") unless 0<=$fa<=1;

my $fb=1-$fa; # fraccion de llenado del aislante.

my $w=zeroes($Nw)->xlinvals($wmin,$wmax); # Arreglo de frecuencias

my $c=1;

my $q=$w/$c; # números de onda en el vacío. Supongo c=1

my $epsa=1-$wp**2/$w**2; # Drude

Figura 34: inicializa2
Calcula la respuesta macroscópica reciclando el código de la �g. 4, y

empléala para obtener la relación de dispersión de super�cie usando las ec.
(20).

my $Q=sqrt(($eM_par-1)*$eM_perp/($eM_par*$eM_perp-1))*$q;

Figura 35: relspp
Gra�ca el resultado.

my $gw=gpwin();

my $S="{/Symbol %s}";

$gw->plot({title=>sprintf("Relación de dispersión SPP ${S}_p=%s ${S}_b=%s f_a=%s f_b=%s",

'w', $wp, 'e', $epsb, $fa, $fb),

xlabel=>'Vector de onda Q', ylabel=>'Frecuencia {/Symbol w}',

xrange=>[undef,$Q1],

},

with=>'lines', $Q, $w,);

prompt -single, -void, "Listo?";

Figura 36: gra�cas
Finalmente, coloca todo junto.

#Calcula la respuesta macroscópica de una superred metal aislante.

<<paquetes>>

<<inicializa2>>

<<calcula>>

<<relspp>>

<<grafica2>>

<<uso>>

Figura 37: todo

3.8. Resultados: Modos de super�cie

Podemos correr el programa como en la �g. 15.

./relspp.pl -wp=1 -epsb=2 -fa=.5 -wmin=.01 -wmax=1 -Nw 1000 -Q1 3

./relspp.pl -wp=1 -epsb=2 -fa=.25 -wmin=.01 -wmax=1 -Nw 1000 -Q1 3

./relspp.pl -wp=1 -epsb=2 -fa=.75 -wmin=.01 -wmax=1 -Nw 1000 -Q1 3

Figura 38: corrida1
La �g. 39 muestra los resultados para el sistema correspondiente a la

�g. 8, con ωp = 1, εb = 2 y fa = fb = 0,5. Los resultados se ven similares
a la relación del plasmón de super�cice en un metal homogéneo, excepto
que el valor asintótico de la frecuencia en el límite no retardado es menor,
correspondiente a un metal menos denso.

La �g. 40 muestra los mismos resultados pero para una menor fracción
de llenado fa = 0,25. La rama inferior muestra una frecuencia asintótica
todavía menor. Curiosamente, aparece una nueva rama con un comporta-
miento peculiar. Sin embargo, ésta se encuentra a la izquierda de la línea de
luz y por tanto no corresponde a modos de super�cie, sino, como se discutió
después de la ec. (20), a la condición de Brewster.

Finalmente, en la �g. 41 se muestran los resultados correspondientes a
una fracción de llenado mayor fa = 0,75. De nueva cuenta, se obtiene una
relación de dispersión similar a la del metal homogéneo, pero con otra fre-
cuencia asintótica.

De acuerdo a la ec. (21) los valores asintóticos del plasmón de super�cie
son ωspp = 0,577ωp, 0,418ωp, y 0,662ωp para fa = 0,5, 0,25 y 0,75 respecti-
vamente, en buen acuerdo con las �guras 39, 40 41.

Figura 39: Relación de dispersión del plasmón de super�cie para el mismo
sistema que en la �g. 8.

Figura 40: Relación de dispersión del plasmón de super�cie como en la �g.
39, pero para fa = 0,25.

Figura 41: Relación de dispersión del plasmón de super�cie como en la �g.
39, pero para fa = 0,75.

4. Conclusiones

Obtuvimos la relación de dispersión de bulto y de super�cie para los
modos electromagnéticos de una superred metal-dieléctrico en el límite de
películas muy delgadas. El sistema es equivalente a un medio homogéneo
anisotrópico. Obtuvimos la respuesta macroscópica, que tiene caracter me-
tálico a lo largo de las interfaces y dieléctrico con una resonancia en la direc-
ción perpendicular. Obtuvimos las frecuencias características que describen
dichas respuestas. La relación de dispersión del bulto tiene varias ramas, co-
rrespondientes a las distintas combinaciones de signo entre las componentes
principales de la respuesta dieléctrica. Sin embargo, los plasmones polari-
tones de super�cie tienen una rama que cualitativamente se ve similar a la
de un metal homogéneo con una densidad electrónica efectiva menor que
la de la fase metálica. Esto sugiere que se pueden entonar los plasmones-
polaritones de super�cie variando la fracción de llenado de las componentes
de una superred, modulando así su velocidad de fase y grupo.

